



# WHAT IS A TREND?

Des Bleakley  
BullCharts User Group  
Tuesday 16<sup>th</sup> March 2021



WHAT IS THIS?



**THIS IS A CLASSIC UPTREND**  
Characterised by Higher Highs  
& Higher Lows (HHHL)





HHHL = UPTREND




WHAT ABOUT THIS ?





HHHL = UPTREND



STILL IN UPTREND = UPTREND  
(CONTINUED)





**HHHL = UPTREND**



**WHAT ABOUT THIS?**





**HHHL = UPTREND**





**STILL IN UPTREND = UPTREND  
(CONTINUED)**



HHHL = UPTREND




WHAT ABOUT THIS -  
UP or DOWN?





**HHHL = UPTREND**



**WHAT ABOUT THIS  
UP or DOWN?**





**STILL IN UPTREND  
(CONTINUED)**

**But somehow doesn't  
look quite right**





HHHL = UPTREND



AND NOW ?




**HHHL = UPTREND**




**AND NOW ?**



**STILL IN UPTREND  
(CONTINUED)**




HHHL = UPTREND




AND NOW ?



**HHHL = UPTREND**



**AND NOW ?**



**NOW WE ARE IN A  
CLASSIC  
DOWNTREND (LHLL)**

**But we might need  
to wait a long time  
for this downtrend  
to kick in**

# SO WHAT?

- ▶ **Does it really matter?**
- ▶ **Does it happen often?**
- ▶ **Is it substantial or just little blips on the trend chart?**
- ▶ **See real world examples**

# REAL WORLD EXAMPLES - Feb 2020



# WHAT TO DO?


- ▶ Sometimes the Classic HHHL is VERY late to change its status and may therefore misrepresent the state of the trend because it needs an extra “kink” to force the change in the HHHL status.
- ▶ Determine Trend has changed when a sufficiently large price movement has taken place (UP & DOWN) - (i.e. - introduce an artificial “kink”)
- ▶ Your choice - I use 7% for a stock (weekly) - less for an Index (set by parameter)
- ▶ Timeframe - weekly vs daily - what % to use
- ▶ Overrides classic HHHL structure
- ▶ Apply in both breakaway UP & DOWN situations
- ▶ Gives a much more rational picture
- ▶ Display as ribbon - choose whichever version you like
  - ▶ CLASSIC / SIMPLE / DETAILED



# EXAMPLE - CBA



# TREND IS CLEAR IRRESPECTIVE OF CHART TYPE



# QUESTIONS

- ▶ BullChart code freely available
- ▶ Contact [des@bleakley.com.au](mailto:des@bleakley.com.au)

Description



Input parameters



Decide Runaway limits



Find Highs & Lows



Classic UP/DOWN?

Runaway?



Previous Trend Continues?



Colour Ribbon Appropriately



```
[description=HH & LL TREND RIBBON - COMES IN THREE VARIANTS (set by parameter)
D - DISPLAY DETAILED TREND CHANGES
C - USE CLASSIC HH & HL
S - SIMPLE UP or DOWN ONLY.
DETAILED AND SIMPLE INCLUDES CODE TO CHANGE TREND WHEN RUNAWAY UP OR DOWN OCCURS.
DEFAULT IS SIMPLE AND RUNAWAY DEFAULTS TO 7%.

COLOURS
RED (DOWN Trend started)
PINK (DOWN Trend continued)
LIME GREEN (UP Trend Started)
YELLOW GREEN (UP Trend continued)
BLACK (UP Trend broken by x%)
DARK GREEN (DOWN Trend broken by x%)
CYAN BLUE (DOWN Trend broken by x% AND UP Trend started).

LAST MODIFIED 9 MAR M00X1 ;author=Bleakley, Des];

num_bars_displayed := input("BARS TO DISPLAY (0=ALL)",75,0,1000);
display_bars := if(num_bars_displayed=0, TRUE, barnumber > lastvalue(barnumber) - num_bars_displayed);
display_type := inputtext("CLASSIC C, SIMPLE S, DETAILED D", "S");
runaway_ratio_pc := input("RUNAWAY %",7,0,100);
display_C := display_type = "C" OR display_type = "c";
display_D := display_type = "D" OR display_type = "d";
display_S := NOT (display_C OR display_D);

{ **** SET RUNAWAY VALUES FROM INPUT PARAMETER **** }

runaway_up_ratio := (100+runaway_ratio_pc)/100;
runaway_down_ratio := (100-runaway_ratio_pc)/100;
earlybar := barnumber<10;

{ **** FIND LAST TWO HIGHS & TWO LOWS (last & previous) **** }

high_point_plus1 := hist(close,2)<hist(close,1) AND hist(close,1)>close;
close_last_high_point := valuewhen(1,high_point_plus1, hist(CLOSE,1)) ;
close_previous_high_point := valuewhen(2,high_point_plus1, hist(CLOSE,1)) ;

low_point_plus1 := hist(close,2)>hist(close,1) AND hist(close,1)<close;
close_last_low_point := valuewhen(1,low_point_plus1, hist(CLOSE,1)) ;
close_previous_low_point := valuewhen(2,low_point_plus1, hist(CLOSE,1)) ;

{ **** DECIDE IF WE ARE HH & HL (UPTREND) OR LH & LL (DOWNTREND) **** }

highs_going_down := close_previous_high_point > close_last_high_point;
lows_going_up := close_previous_low_point < close_last_low_point;
close_higher_than_last_high_point := close_last_high_point < close;
close_lower_than_last_low_point := close_last_low_point > close;

{ **** DECIDE IF WE ARE IN A RUNAWAY TURNAROUND **** }

runaway_up := if(display_C, FALSE, close > close_last_low_point*runaway_up_ratio AND CLOSE > hist(CLOSE,1));
runaway_down := if(display_C, FALSE, close < close_last_high_point*runaway_down_ratio AND CLOSE < hist(CLOSE,1));

{ **** ARE WE IN UP OR DOWN TREND & DOES IT CONTINUE? **** }

UPTR := if(earlybar,FALSE,if( (lows_going_up AND close_higher_than_last_high_point) , TRUE, FALSE));
DNTR := if(earlybar,TRUE,if( (highs_going_down AND close_lower_than_last_low_point) , TRUE,FALSE));
UP_TREND_CONTINUED := if(earlybar,FALSE,if(UPTR OR runaway_up,TRUE,if(DNTR OR runaway_down,FALSE,prev)));
DN_TREND_CONTINUED := if(earlybar,TRUE,if(DNTR OR runaway_down,TRUE,if(UPTR OR runaway_up ,FALSE,prev)));

{ **** DISPLAY COLOURED RIBBONS **** }

[TARGET=RIBBON];
[name=STOCK TREND - H0HL - CLASSIC;color=yellow;]display_C AND NOT display_bars;
[name=STOCK TREND - H0HL - DETAILED;color=yellow;]display_D AND NOT display_bars;
[name=STOCK TREND - H0HL - SIMPLE;color=yellow;]display_S AND NOT display_bars;
[name=UP Classic;color=lime;]display_C AND UPTR;
[name=DN Classic;color=red;]display_C AND DNTR;
[name=UP Simple Trend;color=lime;]display_S AND UP_TREND_CONTINUED;
[name=DN Simple Trend;color=red;]display_S AND DN_TREND_CONTINUED;
[name=UPboth;color=cyan;]display_D AND UPTR AND runaway_up;
[name=DNboth;color=brown;]display_D AND DNTR AND runaway_down;
[name=UPrun;color=green;]display_D AND runaway_up AND NOT hist(runaway_up,1) AND hist(DN_TREND_CONTINUED,1);
[name=DNrun;color=black;]display_D AND runaway_down AND NOT hist(runaway_down,1) AND hist(UP_TREND_CONTINUED,1);
[name=UP Trend;color=lime;]display_D AND UPTR;
[name=DN trend;color=red;]display_D AND DNTR;
[name=UPcon;color=green yellow;]UP_TREND_CONTINUED;
[name=DNcon;color=pink;]DN_TREND_CONTINUED;
[name=ERR;color= yellow;]TRUE;
```